


# State\_Variable\_f\_1Hz

\* dsauersanjose@aol.com  
 \* www.idea2ic.com  
 \* 4/15/08



$$H(s) = 1/(s^2 + s*A1 + A0)$$

$$H(s) = 1/(s^2 + s*w0/Q + w0^2)$$

Set **A0 = 1** and scale **s** to 1KHz  
 Then **R2 = Q** and **s = 2\*PI\*1KHz**


```

.OPTIONS GMIN=1e-18 METHOD=trap srcsteps = 1 gminsteps = 1
=====
V_IN     VIN      0      AC      1      DC      0
R0       VIN      SUMN   1
R1       SUMN    HP      1
R2       SUMN    BP      10
R3       SUMN    LP      1
XOPA1    SUMN    0       HP      OPA
XONES1   HP      BP      ONE_S
XONES2   BP      LP      ONE_S
.ac      dec      50     10     10k
  
```


\*==Scale\_the\_Integrator\_To\_the\_Desired\_Frequency==\*

```

.control
run
plot db(bp) db(hp) db(lp) title StateVariable_Q_10
*====Q_Is_Independant_Of_Frequency====
alter R2 resistance = 1
run
plot db(bp) db(hp) db(lp) title StateVariable_Q_1
.endc
  
```



[http://www.idea2ic.com/PlayWithJavascript/R\\_C\\_Freq.html](http://www.idea2ic.com/PlayWithJavascript/R_C_Freq.html)



```

.SUBCKT OPA INP INN OUT
EGN1 O1 0 INP INN -1
EGN2 OUT 0 IP 0 -1000000
R1 O1 IP 10k
C1 OUT IP 10p
  
```

.ends

```

.SUBCKT ONE_S IN OUT
EGN1 O1 0 IP 0 -1000000
EGN2 OUT 0 O1 0 -1
R1 IN IP 10k
C1 IP O1 .01592u
.ends


```

.end

=====END\_OF\_SPICE=====

Scaling frequency involves adjusting the R1 value and C1 value to equal each other at the desired frequency inside the integrator block.

Graph 104 - ac314: StateVariable\_Q\_1



The External Resistors R1->R3 indendently control Q

Graph 103 - ac313: StateVariable\_Q\_10

